" Desc: Converts Youtube transcripts to srt
" Input: 
"   0:00
"   text
"   0:06
"   text
"   0:12
"   text
" Output: 
"   1
"   00:00:00,000 --> 00:00:06,000
"   text
"   2
"   00:00:06,000 --> 00:00:12,000
"   text
"   3
"   00:00:12,000 -->
"   text

" ^0:10 ==> ^00:10
%s/^\ze\d:/0/

" ^00:10 ==> 00:00:10
%s/\v^\ze\d+(:\d+)$/00:/g

" 00:00:10 ==> 00:00:10,000
%s/\v\d+(:\d+)+\zs/,000/g

" ^00:00:10 ==> ^00:00:10 --> 00:00:30
g/\v^\d+(:\d+)+(,\d+)$/ exec "norm A --> " .. getline(line(".")+2)

let i=1
" ^00:00:10 --> 00:00:30 ==>
" ^1
" ^00:00:10 --> 00:00:30
g/\v^\d+(:\d+)+/ exec "let i+=1 | norm O" .. i
